Ellipsometry

Introduction

Ellipsometry is the measurement of the effect of reflection on the state of polarization of light. The result of an ellipsometric measurement can be the complex refractive index of the reflecting material, or if the reflecting material is a film-covered substrate, the thickness and optical constants of the film can be determined. Ellipsometry is particularly attractive because it does not perturb the sample being measured and it is extremely sensitive to minute interfacial effects and can be applied to surface films having a thickness as small as monoatomic to as large as several microns. Any substrate-film-ambient combination that provides reasonably specular reflection of the incident light beam can be measured. Scattering during the reflection process causes partial depolarization of the incident beam and, consequently, reduced precision and accuracy.

Since ellipsometry essentially measures the state of polarization of reflected or transmitted light it can be thought of as polarimetry. The state of polarization is defined by the phase and amplitude relationships between the two component plane waves into which the electric field is resolved. The wave having the electric field in the plane of incidence is called the p wave, and the wave having the electric field normal to the plane of incidence is called the s wave. If the p and s components are in phase, or 180 degrees out of phase, the resultant wave is plane polarized. A difference of phase, other than 180°, corresponds to elliptical polarization. In general, reflection causes a change in relative phases of the p and s waves and a change in the ratio of their amplitudes. The change in phase is characterized by the angle Δ, and the amplitude ratio change is characterized by Tan[ψ]. If the amplitudes of the incident and reflected beams are designated e and r, respectively, and phases of the incident and reflected beams are α and β, respectively

$$\mathrm{Tan}[\psi] = \frac{|r_p|}{|r_s|} \frac{|e_s|}{|e_p|}$$

$$\Delta = (\beta_p - \beta_s) - (\alpha_p - \alpha_s)$$

Ellipsometry is the measurement of ψ and Δ.

Measurement Principles

The principle of the measurement of ψ and Δ is explained with the help of Figure 1, which is a schematic representation of an ellipsometer. The incident monochromatic beam is collimated and transmitted through a linear polarizer and compensator (retarder). (In some ellipsometers a broad spectral band source is used and ψ and Δ are measured as a function of wavelength. In this discussion we will consider only monochromatic illumination.) The azimuthal orientations of the polarizer and compensator determine the relative amplitudes and phase difference between the p and s components of the beam incident upon the substrate. These orientations are adjusted so the difference in phase just compensates that which results from reflection off the sample. The plane polarized beam reflected off the sample is transmitted by the analyzer to a telescope and detector and the analyzer is oriented to extinguish the reflected beam. Δ and ψ are determined from the orientation of the polarizer and analyzer for extinction.



Figure 1. Schematic representation of ellipsometer.

In the discussion below it will be assumed that the polarizer, analyzer, and compensator are ideal. It is assumed that the compensator is a wave-plate introducing a retardation of δ and no attenuation. The orientation of the wave-plate is selected so the slow axis is inclined at 45° to the plane of incidence. Any angle can be used, but the compensator is generally used at ± 45°. Let p be the angle between the polarizer transmission axis and the x-axis which is taken to be the direction for p polarization as illustrated in Figure 2.



Figure 2. Orientation of polarizer, compensator, and analyzer.

The light transmitted through the polarizer can be written in the form of a Jones vector as

In[1]:=
```
lightLinear = (Cos[p]
               Sin[p]);
```

A retarder with the fast axis horizontal can be written in terms of a Jones Matrix as

In[2]:= $\texttt{rfah}[\delta\_] := e^{-i\,\delta/2}\begin{pmatrix}1 & 0\\ 0 & e^{i\,\delta}\end{pmatrix}$

A rotation matrix can be written as

In[3]:= $\texttt{rot}[\theta\_] := \begin{pmatrix}\text{Cos}[\theta] & \text{Sin}[\theta]\\ -\text{Sin}[\theta] & \text{Cos}[\theta]\end{pmatrix}$

A retarder of retardation δ having a fast axis at an angle of θ from the horizontal can be written as

```
In[4]:= rrot[δ_, θ_] := rot[-θ].rfah[δ].rot[θ]
```

Thus, the light transmitted through the wave plate and incident upon the sample can be written as

```
In[94]:= lightIncident = FullSimplify[rrot[δ, -45 °].lightLinear] // MatrixForm;
```

The p-component (x) can be written as

In[6]:= $\texttt{pComponent} = \text{Cos}[p]\,\text{Cos}\left[\frac{\delta}{2}\right] + i\,\text{Sin}[p]\,\text{Sin}\left[\frac{\delta}{2}\right];$

and the s-component (y) can be written as

$\texttt{sComponent} = \text{Cos}\left[\frac{\delta}{2}\right]\text{Sin}[p] + i\,\text{Cos}[p]\,\text{Sin}\left[\frac{\delta}{2}\right];$

■ Phase determination

The tangent of the phase of the p component can be written as

In[8]:= $\texttt{tanpComponent} = \frac{\text{Sin}[p]\,\text{Sin}\left[\frac{\delta}{2}\right]}{\text{Cos}[p]\,\text{Cos}\left[\frac{\delta}{2}\right]};$

The tangent of the phase of the s component can be written as

In[9]:= $\texttt{tansComponent} = \frac{\text{Cos}[p]\,\text{Sin}\left[\frac{\delta}{2}\right]}{\text{Cos}\left[\frac{\delta}{2}\right]\text{Sin}[p]};$

The goal is to find the tangent of the phase difference between the p and s components. Remembering that

$$\text{Simplify}\left[\frac{\text{Tan}[\alpha]-\text{Tan}[\beta]}{1+\text{Tan}[\alpha]\,\text{Tan}[\beta]}\right] = \text{Tan}[\alpha-\beta]$$

we can write

$\texttt{tan}\Delta\texttt{Incident} =$

$$\text{Factor}\left[\text{TrigExpand}\left[\text{FullSimplify}\left[\frac{\frac{\text{Sin}[p]\,\text{Sin}\left[\frac{\delta}{2}\right]}{\text{Cos}[p]\,\text{Cos}\left[\frac{\delta}{2}\right]} - \frac{\text{Cos}[p]\,\text{Sin}\left[\frac{\delta}{2}\right]}{\text{Cos}\left[\frac{\delta}{2}\right]\text{Sin}[p]}}{1+\frac{\text{Sin}[p]\,\text{Sin}\left[\frac{\delta}{2}\right]}{\text{Cos}[p]\,\text{Cos}\left[\frac{\delta}{2}\right]}\,\frac{\text{Cos}[p]\,\text{Sin}\left[\frac{\delta}{2}\right]}{\text{Cos}\left[\frac{\delta}{2}\right]\text{Sin}[p]}}\right]\right]\right] = -\frac{1}{2}\,\text{Sin}[\delta]\,(\text{Cot}[p]-\text{Tan}[p])$$

But,

$$\text{FullSimplify}\left[\text{TrigToExp}\left[-\frac{1}{2}\,(\text{Cot}[p]-\text{Tan}[p])\right]\right] = -\text{Cot}[2\,p]$$

Furthermore,

$$\text{Tan}[2\,p-90\,°] = -\text{Cot}[2\,p]$$

Therefore,

```
tanΔIncident = Sin[δ] Tan[2 p - 90 °];
```

It is interesting to look at a plot of Δ as a function of p. To correct for a discontinuity in the ArcTan function 180 degrees will be added for p≥ 0 to make the function continuous.

Out[93]=



A 180 degree rotation of the polarizer introduces a 360 degree change in the retardation. If the compensator is a quarter-wave plate ($\delta=\pi/2$) there is a linear relationship between ΔIncident, the phase difference between the p and s components of the light incident upon the sample, and p, the orientation angle of the polarizer.

Amplitude determination

Next we will look at the ratio of the amplitudes of the s and p components of the electric field incident upon the sample. Let

$$\text{tanL} = \frac{|e_p|}{|e_s|}$$

Then

```
tanLSquared = ComplexExpand[pComponent Conjugate[pComponent]] / ComplexExpand[sComponent Conjugate[sComponent]];
```

Remembering that

$$\text{Simplify}\left[\frac{1 - \text{Tan}[\alpha]^2}{1 + \text{Tan}[\alpha]^2}\right] = \text{Cos}[2\,\alpha]$$

$$\text{cos2L} = \text{Simplify}\left[\frac{1 - \text{tanLSquared}}{1 + \text{tanLSquared}}\right] = -\text{Cos}[2\,p]\,\text{Cos}[\delta];$$

Similar relationships are obtained with appropriate changes in sign if the slow axis is oriented at -45° to the plane of incidence.

It is interesting to note that if the compensator is a quarter-wave plate the orientation of the polarizer has no effect upon the ratio of the amplitudes of the s and p components of the electric field incident upon the sample.

Measurement procedure

The measuring procedure consists of adjusting the polarizer and analyzer so the detected beam is extinguished. There are two orientations of the polarizer which lead to plane polarized light. The two conditions are

$$\Delta\text{Incident} = -\Delta\text{Sample}$$

and

$$\Delta\text{Incident} = -\Delta\text{Sample} + 180°$$

It follows from the equation for TanΔIncident that the two conditions for plane polarized light being reflected off the sample are

$$\tan\Delta\text{Sample} = \text{Sin}[\delta]\ \text{Tan}[90° - 2 p_1]$$

and

$$\tan\Delta\text{Sample} = \text{Sin}[\delta]\ \text{Tan}[270° - 2 p_2]$$

At extinction, the analyzer transmission axis orientation angle, a, is equal to r ± 90°, where r is the angle of the reflected linear polarization relative to the plane of incidence.

$$\text{Tan}[r] = \frac{|r_s|}{|r_p|}$$

$$\text{Tan}[\psi] = \frac{|r_p|}{|r_s|}\ \frac{|e_s|}{|e_p|}$$

$$\text{Tan}[\psi] = \frac{\text{Tan}[-a_1]}{\text{Tan}[L_1]}$$

and for the second set of angles

$$\text{Tan}[\psi] = \frac{\text{Tan}[a_2]}{\text{Tan}[L_2]}$$

Since $\text{Cot}[L_1] = \text{Tan}[L_2]$

$$\text{Tan}[\psi]^2 = \text{Tan}[-a_1]\ \text{Tan}[a_2]$$

If the compensator is a quarter-wave plate, $\delta = 90°$, the relationships between Δ and ψ and the extinction settings are especially simple.

$$\Delta = 90° - 2 p_1 = 270° - 2 p_2$$

$$\psi = -a_1 = a_2.$$

Interpretation of data

Using the measured values of Δ and ψ it is possible to determine the complex refractive index of substrates and the thickness and refractive index of thin films, however the equations are extremely complicated and their solution and use for interpretating ellipsometric data requires electronic computation. Details on the specific computations are beyond the scope of these notes and they can be found in reference 1.

References

1) Azzam, R.M.A. and Bashara, N.M., (1988). "Ellipsometry and Polarized Light", North-Holland, New York.
2) Archer, R.J. "Manual on Ellipsometry", Gaertner Scientific, Skokie, IL.
3) Spanier, R., (September, 1975). "Ellipsometry, A Century Old New Science", Industrial Research.
4) Hecht, E., (1998). "Optics", Addison Wesley, New York.
5) Born, M. and Wolf, E., (1959). "Principles of Optics", Pergamon Press, New York.

Fresnel Equations

Bare substrates

Fresnel first derived equations for the reflection coefficients of bare surfaces in terms of the angle of incidence, angle of refraction, and the complex refractive index. The results for the amplitude reflection coefficient and amplitude transmission coefficient are given below. The sign convention used is not standarized. For our equations the sign convention used in reference 4 (Hecht) is followed.

$$r_s = \frac{n_i \,\mathrm{Cos}[\theta_i] - n_t \,\mathrm{Cos}[\theta_t]}{n_i \,\mathrm{Cos}[\theta_i] + n_t \,\mathrm{Cos}[\theta_t]}$$

$$t_s = \frac{2\, n_i \,\mathrm{Cos}[\theta_i]}{n_i \,\mathrm{Cos}[\theta_i] + n_t \,\mathrm{Cos}[\theta_t]}$$

$$r_p = \frac{n_t \,\mathrm{Cos}[\theta_i] - n_i \,\mathrm{Cos}[\theta_t]}{n_i \,\mathrm{Cos}[\theta_t] + n_t \,\mathrm{Cos}[\theta_i]}$$

$$t_p = \frac{2\, n_i \,\mathrm{Cos}[\theta_i]}{n_i \,\mathrm{Cos}[\theta_t] + n_t \,\mathrm{Cos}[\theta_i]}$$

For an optically absorbing medium the complex index of refraction of the substrate is given by

$$n_t = n - i\, k;$$

From the definitions given above it follows that

$$\frac{r_p}{r_s} = \mathrm{Tan}[\psi]\, e^{i\,\Delta}$$

The algebra for solving for n and k from ψ and Δ is extremely messy and will not be given here. The details can be found in references 1 and 2.

Phase change at normal incidence

$$r = \frac{1 - n_t}{1 + n_t}$$

$$\frac{1 + i\, k - n}{1 - i\, k + n}$$

```
rBottom = Simplify[Denominator[r] Conjugate[Denominator[r]], {k > 0, n > 0}]
```

$$k^2 + (1 + n)^2$$

```
rTop = Simplify[Numerator[r] Conjugate[Denominator[r]], {k > 0, n > 0}]
```

$$1 + 2\, i\, k - k^2 - n^2$$

$$\text{phase}[n\_, k\_] := \text{ArcTan}\left[\frac{2\, k}{1 - n^2 - k^2}\right]$$

Thin films

The reflectance of substrates having a coating of thin films can be calculated using the characteristic matrix approach as described in references 4 and 5. Δ and ψ can be calculated in terms of the angle of incidence, the wavelength, the optical constants of the film and substrate and the thickness of the film. The equations are extremely complicated and their solution and use for interpretating ellipsometric data requires electronic computation. References 1 and 2 give additional information.

References

1) Azzam, R.M.A. and Bashara, N.M., (1988). "Ellipsometry and Polarized Light", North-Holland, New York.
2) Archer, R.J. "Manual on Ellipsometry", Gaertner Scientific, Skokie, IL.
3) Spanier, R., (September, 1975). "Ellipsometry, A Century Old New Science", Industrial Research.
4) Hecht, E., (1998). "Optics", Addison Wesley, New York.
5) Born, M. and Wolf, E., (1959). "Principles of Optics", Pergamon Press, New York.

Measuring Birefringence

Illuminate sample with circularly polarized light. Put rotating analyzer between sample and detector and measure light transmitted thru sample and analyzer.



Basic Definitions

- **Circular polarization**

$$\texttt{stokes} = \begin{pmatrix} 1 \\ 0 \\ 0 \\ 1 \end{pmatrix};$$

- **Horizontal linear polarizer**

$$\texttt{hlpMueller} = \frac{1}{2}\begin{pmatrix} 1 & 1 & 0 & 0 \\ 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix};$$

- **Linear retarder of retardation δ with fast axis horizontal**

$$\texttt{retarderHorizontal}[\delta\_] := \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \texttt{Cos}[\delta] & \texttt{Sin}[\delta] \\ 0 & 0 & -\texttt{Sin}[\delta] & \texttt{Cos}[\delta] \end{pmatrix}$$

- **Rotation Matrix**

$$\texttt{rotMueller}[\theta\_] := \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \texttt{Cos}[2\,\theta] & \texttt{Sin}[2\,\theta] & 0 \\ 0 & -\texttt{Sin}[2\,\theta] & \texttt{Cos}[2\,\theta] & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

- **Calculation of matrix of a retarder of retardation δ having a fast axis at an angle θ from the horizontal**

```
rrot[δ_, θ_] := rotMueller[-θ].retarderHorizontal[δ].rotMueller[θ]
```

Measuring the birefringence

Rotate polarizer in front of detector and determine the birefringence δ and the angle of the birefringence, θ. ω t is the angle of the polarizer.

- **Measure phase and amplitude of signal as polarizer rotates**

```
polarimeter = rotMueller[-ω t].hlpMueller.rotMueller[ω t].rrot[δ, θ].stokes;
```

```
polarimeter[[1]][[1]] // Simplify
```

$$\frac{1}{4}\left(2+\text{Cos}[\delta+2\theta-2t\omega]-\text{Cos}[\delta-2\theta+2t\omega]\right)$$

```
signal = 1/4 (2 + Simplify[Cos[δ + 2 θ - 2 t ω] - Cos[δ - 2 θ + 2 t ω]]) // Simplify
```

$$\frac{1}{2}\left(1-\text{Sin}[\delta]\,\text{Sin}[2(\theta-t\omega)]\right)$$

The amplitude of the Sin[2ωt] signal is given by the Sin[birefringence] and the phase of the signal is given by 2 times the angle of the birefringence, θ.

- **Measure signal for discrete positions of polarizer**

```
signal1 = signal /. t ω -> 0
```

$$\frac{1}{2}\left(1-\text{Sin}[\delta]\,\text{Sin}[2\theta]\right)$$

```
signal2 = signal /. t ω -> π / 4 // FullSimplify
```

$$\frac{1}{2}\left(1+\text{Cos}[2\theta]\,\text{Sin}[\delta]\right)$$

```
signal3 = signal /.  t ω -> π / 2
```

$$\frac{1}{2}\left(1-\text{Sin}[\delta]\,\text{Sin}\left[2\left(-\frac{\pi}{2}+\theta\right)\right]\right)$$

```
signal3 = signal3 /. Sin[2 (-π/2 + θ)] -> TrigReduce[Sin[2 (-π/2 + θ)]]
```

$$\frac{1}{2}\left(1+\text{Sin}[\delta]\,\text{Sin}[2\theta]\right)$$

```
signal4 = signal /. t ω -> 3 π / 4 // Simplify
```

$$\frac{1}{2}\left(1-\text{Cos}[2\theta]\,\text{Sin}[\delta]\right)$$

```
(signal3 - signal1)/(signal2 - signal4) // Simplify
```

$$\text{Tan}[2\theta]$$

If we take an ArcTan we obtain the orientation of the birefringence.

```
(signal3 - signal1) // Simplify
```

$$\text{Sin}[\delta]\,\text{Sin}[2\theta]$$

```
(signal2 - signal4) // Simplify
```

$$\text{Cos}[2\theta]\,\text{Sin}[\delta]$$

```
(signal3 - signal1)^2 + (signal2 - signal4)^2 // Simplify
```

```
Sin[δ]^2
```

If we take the ArcSin of the square root we get the magnitude of birefringence. Since we know 2θ, the sign of (signal3-signal1) gives us the sign of the birefringence.

Measuring the Stokes Parameters

One method for measuring the Stokes parameters is to measure the intensity of the light after it passes through a rotating quarter-wave plate followed by a horizontal linear polarizer. The following four intensity measurements are required:

Fast-axis of the quarter-wave plate at

a) 0°,
b) 30°,
c) 60°, and
d) 135°.

Basic Definitions

Stokes Vector

$$\mathbf{stokes} = \begin{pmatrix} s0 \\ s1 \\ s2 \\ s3 \end{pmatrix};$$

Horizontal linear polarizer

$$\mathbf{hlpMueller} = \frac{1}{2}\begin{pmatrix} 1 & 1 & 0 & 0 \\ 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix};$$

Linear retarder of retardation δ with fast axis horizontal

$$\mathbf{retarderHorizontal}[\delta\_] := \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \mathrm{Cos}[\delta] & \mathrm{Sin}[\delta] \\ 0 & 0 & -\mathrm{Sin}[\delta] & \mathrm{Cos}[\delta] \end{pmatrix}$$

Rotation Matrix

$$\mathbf{rotMueller}[\theta\_] := \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \mathrm{Cos}[2\,\theta] & \mathrm{Sin}[2\,\theta] & 0 \\ 0 & -\mathrm{Sin}[2\,\theta] & \mathrm{Cos}[2\,\theta] & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

Quarter-wave plate at angle θ

```
qwp[θ_] := rotMueller[-θ].retarderHorizontal[π/2].rotMueller[θ]
```

Polarimeter Output

Rotating quarter-wave plate and horizontal linear polarizer

Fast axis of quarter-wave plate at 0°.

```
output1 = hlpMueller.qwp[0].stokes; MatrixForm[output1]
```

$$\begin{pmatrix} \frac{s0}{2} + \frac{s1}{2} \\ \frac{s0}{2} + \frac{s1}{2} \\ 0 \\ 0 \end{pmatrix}$$

Fast axis of quarter-wave plate at 30°.

```
output2 = hlpMueller.qwp[π/6].stokes; MatrixForm[output2]
```

$$\begin{pmatrix} \frac{s0}{2} + \frac{s1}{8} + \frac{\sqrt{3}\ s2}{8} - \frac{\sqrt{3}\ s3}{4} \\ \frac{s0}{2} + \frac{s1}{8} + \frac{\sqrt{3}\ s2}{8} - \frac{\sqrt{3}\ s3}{4} \\ 0 \\ 0 \end{pmatrix}$$

Fast axis of quarter-wave plate at 60°.

```
output3 = hlpMueller.qwp[π/3].stokes; MatrixForm[output3]
```

$$\begin{pmatrix} \frac{s0}{2} + \frac{s1}{8} - \frac{\sqrt{3}\ s2}{8} - \frac{\sqrt{3}\ s3}{4} \\ \frac{s0}{2} + \frac{s1}{8} - \frac{\sqrt{3}\ s2}{8} - \frac{\sqrt{3}\ s3}{4} \\ 0 \\ 0 \end{pmatrix}$$

Fast axis of quarter-wave plate at 135°.

```
output4 = hlpMueller.qwp[3/4 π].stokes; MatrixForm[output4]
```

$$\begin{pmatrix} \frac{s0}{2} + \frac{s3}{2} \\ \frac{s0}{2} + \frac{s3}{2} \\ 0 \\ 0 \end{pmatrix}$$

The four Stokes parameters are calculated as follows:

```
ans = Solve[{output1[[1, 1]] == reading1, output2[[1, 1]] == reading2,
    output3[[1, 1]] == reading3, output4[[1, 1]] == reading4}, {s0, s1, s2, s3}];
{s0 /. ans, s1 /. ans, s2 /. ans, s3 /. ans} // MatrixForm
```

$$\begin{pmatrix} \frac{2\left(-\sqrt{3}\ \text{reading1}+2\sqrt{3}\ \text{reading2}+2\sqrt{3}\ \text{reading3}+6\ \text{reading4}\right)}{3\left(2+\sqrt{3}\right)} \\ \frac{4\left(3\ \text{reading1}+2\sqrt{3}\ \text{reading1}-\sqrt{3}\ \text{reading2}-\sqrt{3}\ \text{reading3}-3\ \text{reading4}\right)}{3\left(2+\sqrt{3}\right)} \\ \frac{4}{3}\left(\sqrt{3}\ \text{reading2}-\sqrt{3}\ \text{reading3}\right) \\ \frac{2\left(\sqrt{3}\ \text{reading1}-2\sqrt{3}\ \text{reading2}-2\sqrt{3}\ \text{reading3}+3\sqrt{3}\ \text{reading4}\right)}{3\left(2+\sqrt{3}\right)} \end{pmatrix}$$